
Given definitional axioms:

(equal (tlp x)
       (if (endp x)
          (equal x nil)
        (tlp (cdr x))))

(equal (addtoend e l)
       (app l (cons e nil)))

(equal (integerlistp x)
       (if (endp x)
           (equal x nil)
          (and (integerp (car x))
               (integerlistp (cdr x)))))

Let us first prove

  (tlp (addtoend e l))

using the lemma tlp-app

  (implies (tlp y)
           (tlp (app x y)))

This isn't an equality proof, so we need to make a series of reductions:

  (tlp (addtoend e l))
<= { Def. addtoend }
  (tlp (app l (cons e nil)))
<= { Instantiation: Lemma tlpapp }
  (tlp (cons e nil))
<= { Def. tlp, IFfalse axiom, endp def., conspcons axiom }
  (tlp (cdr (cons e nil)))
<= { cdrcons axiom }
  (tlp nil)
= { Evaluation }
  t

Basically, using a propositional deduction on (tlp (cons e nil)) and
an instantiation of the lemma,

   (implies (tlp (cons e nil))
           (tlp (app l (cons e nil))))

allowed me to conclude (tlp (app l (cons e nil))).  In fact, the
particular propositional deduction is Modus Ponens.



What if I said to prove (consp 42) using the lemma (integerp nil)?
Believe it or not, I can construct such a proof:

  (consp 42)
<= { Propositional deduction }
  nil
<= { Evaluation }
  (integerp nil)
<= { Lemma }
  t

What is interesting is that the assumption that (integerp nil) is a lemma
stipulates that it is a theorem.  In this case, the lemma given was not
a theorem; in fact, it always evaluates to nil!  This allowed us--through
legal deductions on flawed assumptions--to conclude nil is a theorem.  Once
we have concluded nil, or "false", we can use a propositional deduction to
conclude anything.  (false -> p  is a tautology.)

Is this really a proof?  Yes it is, but not with regard to any theory ACL2
will allow.  If an extension of ACL2's base theory is able to prove
(integerp nil) then that extension is unsound, because it allows us to
conclude two contradictary propositions: (not (integerp nil)) and
(integerp nil).

ACL2 does not allow unsound extensions of its theory, so this will not happen
if the Lemmas you are given are actually theorems in a proper extension of
ACL2's base theory.

(It's not important that you completely understand soundness yet.)

========================================================================

Prove

  (implies (and (integerlistp l)
                (integerp e))
           (integerlistp (addtoend e l)))

using the lemma integer-listp-app:

  (implies (and (integerlistp x)
                (integerlistp y))
           (integerlistp (app x y)))



Once again, we are not proving an equality, but this time we have assumptions
we can use, we will put it in the context box:

Context:  

  

(integerlistp (addtoend e l))
<= { Defn addtoend }
  (integerlistp (app l (cons e nil)))
<= { Lemma Integerlistpapp }
  (and (integerlistp l)
       (integerlistp (cons e nil)))
<= { Prop. deduction, assumption 1 }
  (integerlistp (cons e nil))
<= { Defn integerlistp, IF axiom, not endp cons }
  (and (integerp (car (cons e nil)))
       (integerlistp (cdr (cons e nil))))
<= { carcons and cdrcons }
  (and (integerp e)
       (integerlistp nil))
<= { Prop. deduction with Assumption 2 }
  (integerlistp nil)
<= { Evaluation }
  t

A proof displaying Case-analysis:

Given the function definitions:

(defun integerlistp (x)
  (if (endp x)
    (equal x nil)
    (and (integerp (car x))
         (integerlistp (cdr x)))))

(defun keepints (l)
  (if (endp l)
    nil
    (if (integerp (car l))
      (cons (car l)
            (keepints (cdr l)))
      (keepints (cdr l)))))

Assumption1:(integerplistp l)
Assumption2: (integerp e)



Let us prove some cases of

  (integerlistp (keepints x))

Let us start with

  (implies (endp x)
           (integerlistp (keepints x)))

  (integerlistp (keepints x))
<= { Def keepints, A1, iftrue axiom }
  (integerlistp nil)
<= { Evaluation }
  t

===========================================================

Let us next prove

  (implies (and (consp x)
               (integerlistp (keepints (cdr x))))
           (integerlistp (keepints x)))

  (integerlistp (keepints x))
<= { Def keepints, assumption 1, iffalse axiom}
  (integerlistp (if (integerp (car x))
                   (cons (car x)
                         (keepints (cdr x)))
                   (keepints (cdr x))))

Now there does not seem to be much we can do here, because (car x) could
be an integer or it could be something else.  How do we prove it holds in
both cases?

Context
A1: (endp x)

Context
Assumption1: (consp x)
Assumption2: (integerlistp (keepints (cdr x)))



One of our rules of inference can help us out here.  A kind of propositional
deduction is case analysis, which comes from the tautology

  ((p -> q) /\ (~p -> q)) -> q

That says that if q is true when we assume p and also true when we assume
~p, then q is always true.  That means we can reduce proving any proposition
q to proving p -> q and ~p -> q for any formula p.  This is a case of
propositional deduction.

For the problem we are working on, let us first assume (integerp (car x))
and prove the conjecture, and then we will assume (not (integerp (car x)))
and prove the conjecture.  That will complete the proof of the original
conjecture. This splits my proof into two cases(subgoals) since there is a
conjunction in ((p -> q) /\ (~p -> q))!

Case1:  (integerp (car x)) is true, so we update our context box for only this sub-proof:

  (integerlistp (if (integerp (car x))
                   (cons (car x)
                         (keepints (cdr x)))
                   (keepints (cdr x))))
<= { CaseAssumption1, iftrue }
  (integerlistp (cons (car x)
                       (keepints (cdr x))))
<= { Def integerlistp, conspcons, endp def, iffalse }
  (and (integerp (car (cons (car x)
                            (keepints (cdr x)))))
       (integerlistp (cdr (cons (car x)
                                 (keepints (cdr x))))))
<= { carcons, cdrcons axioms}
  (and (integerp (car x))
       (integerlistp (keepints (cdr x))))
<= { CaseAssumption1, Assumption2, propositional deduction }
  t

That completes proof of first case(subgoal).

Context(Case1)
Assumption1: (consp x)
Assumption2: (integerlistp (keepints (cdr x)))
CaseAssumption1: (integerp (car x))



Case 2:  (not (integerp (car x)))

  (integerlistp (if (integerp (car x))
                   (cons (car x)
                         (keepints (cdr x)))
                   (keepints (cdr x))))
<= { CaseAssumption2, iffalse }
  (integerlistp (keepints (cdr x)))
<= { Assumption2 }
  t

That completes proof of second case(subgoal).
Having proven both cases, the proof is complete.

Context(Case2)
Assumption1: (consp x)
Assumption2: (integerlistp (keepints (cdr x)))
CaseAssumption2: (not (integerp (car x)))


